BANGALORE UNIVERSITY

LAB MANUAL

VI SEMESTER – PAPER VII

CBSE SCHEME WITH EFFECT FROM 2014-2015

Practical Paper-VII

List of Programs

1. i)Expressing a vector as a linear combination of given set of vectors.
   ii)Examples on linear dependence and independence of vectors.
2. i)Basis and dimension.
   ii)verifying whether a given transformation is linear.
3. Finding a matrix of linear transformation.
4. Finding a linear transformation of a matrix.
5. Verification of Rank-Nullity theorem.
6. Solutions to problems on total differential equations.
7. Solutions to the problems on different types of partial differential equations (Type1 &Type2)
8. Solutions to the problems on different types of partial differential equations (Type3 &Type4)
9. Solving second order linear partial differential equations in two variables with constant co-efficients.
10. Solving second order linear partial differential equations in two variables with constant co-efficients.
11. Solution of one dimensional heat equation using Fourier series with Dirichlet condition.
12. Solution of one dimensional wave equation using Fourier series with Dirichlet condition.

Lab1:**<u>Expressing a vector as a linear combination of given set of vectors.</u>**

1)Express the vector (1,7,-4) as a linear combination of the vectors (1,-3,2) and (2,-1,1).

Program:

```
kill(all)$
v:[1,7,-4]$
v1:[1,-3,2]$
v2:[2,-1,1]$
e:c1*v1+c2*v2$
[globalsolve:true,programmode:true]$
h:linsolve([e[1]=v[1],e[2]=v[2],e[3]=v[3]],[c1,c2]);
V:c1*v1+c2*v2$
if(V=v)then
print("The linear combination of vector",v,"=",c1,v1,"+",c2,v2)
else
print("v is not a linear combination of v1,v2")$
```

Output:

```
solve: dependent equations eliminated: (3)
(%o6) [c1:-3,c2:2]
The linear combination of vector [1,7,-4]=-3[1,-3,2]+2[2,-1,1]
```

Problems:

2) Express (3,7,-4) as a linear combination of the vectors (1,2,3),(2,3,7) and (3,5,6).

3) Express the vector (2,-5,4) as a linear combination of the vectors (1,-3,2) and (2,-1,1).

4) Express the vector (3,-1,1,-2) as a linear combination of the vectors (1,1,0,-1), (1,1,-1,0) and (1,-1,0,0).

Lab2:**<u>Examples on linear dependence and independence of vectors.</u>**

1) Show that the vectors (1,1,-1), (2,-3,5) and (-2,1,4) of R^3 are linearly dependent.

Program:

```
kill(all)$
A:matrix([1,1,-1],[2,-3,5],[-2,1,4]);
D:determinant(A);
if(D=0)then
```

```
disp("The given vectors are Linearly Dependent")
else
disp("The given vectors are Linearly Independent")$
```

Output:

(%o0) *done*

(%o1) $\begin{bmatrix} 1 & 1 & -1 \\ 2 & -3 & 5 \\ -2 & 1 & 4 \end{bmatrix}$

(%o2) -31

The given vectors are Linearly Independent

2) Examine whether the vectors (1,√2,1), (1,-√2,1) and (-1,0,1) of V3(R) are linearly independent.

   Program:

```
kill(all)$
B:matrix([1,sqrt(2),1],[1,-sqrt(2),1],[-1,0,1]);
D:determinant(B);
if (D=0)then
disp("THE GIVEN VECTORS ARE LINEARLY DEPENDENT")
else
disp("THE GIVEN VECTORS ARE LINEARLY INDEPENDENT")$
```

   Output:

(%o1) $\begin{bmatrix} 1 & \sqrt{2} & 1 \\ 1 & -\sqrt{2} & 1 \\ -1 & 0 & 1 \end{bmatrix}$

(%o2) $-2^{5/2}$

THE GIVEN VECTORS ARE LINEARLY INDEPENDENT

3)Examine whether the set of vectors {(1,1,2), (-3,1,0), (1,-1,1), (1,2,-3)} are linearly independent or linearly dependent.

Program:

```
kill(all)$
v1:[1,1,2]$
v2:[-3,1,0]$
```

```
v3:[1,-1,1]$
v4:[1,2,-3]$
e:c1*v1+c2*v2+c3*v3+c4*v4$
h:linsolve([e[1]=0,e[2]=0,e[3]=0],[c1,c2,c3,c4]);
if((c1=0)and (c2=0) and (c3=0)and(c4=0) then
disp("THE GIVEN VECTORS ARE LINEARLY INDEPENDENT")
else
disp("THE GIVEN VECTORS ARE LINEARLY DEPENDENT")$
```

Output:

(%o6) $[c1=-\frac{\%r1}{8}, c2=\frac{11\ \%r1}{8}, c3=\frac{13\ \%r1}{4}, c4=\%r1]$

THE GIVEN VECTORS ARE LINEARLY DEPENDENT

4) Examine whether the set of vectors
{(1,2,1,2), (3,2,3,2),(-1,-3,0,4),(0,4,-1,-3)} are linearly dependent.

Lab3: **Basis and dimension**.

1) Show that the vectors (1,1,0), (0,1,0), (0,1,1) form a basis of R3

Program:

```
kill(all)$
A:matrix([1,1,0],[0,1,0],[0,1,1]);
D:determinant(A);
if(D=0)then
disp("The given vectors are Linearly Dependent and hence not a
basis.")
else
disp("The given vectors are Linearly Independent and hence form a
basis.")$
```

Output:

(%o1) $\begin{bmatrix} 1 & 1 & 0 \\ 0 & 1 & 0 \\ 0 & 1 & 1 \end{bmatrix}$

(%o2) 1

The given vectors are Linearly Independent and hence form a basis.

2) Show that the set of vectors {(1,2,3),(3,1,0),(-2,1,3)} is not a basis of R^3.Determine the dimension and basis of the subspace spanned by the given vectors.

Program:

```
kill(all)$
A:matrix([1,2,3],[3,1,0],[-2,1,3]);
D:determinant(A);
if(D=0)then
disp("The given vectors are Linearly Dependent and hence not a basis")
else
disp("The given vectors are Linearly Independent and hence a basis")$
e:echelon(A);
print("The vectors ",e[1],",",e[2],"are the basis of the subspace")$
r:rank(A)$
print("Dimension of the subspace is=",r)$
```

Output:

(%o1) $\begin{bmatrix} 1 & 2 & 3 \\ 3 & 1 & 0 \\ -2 & 1 & 3 \end{bmatrix}$

(%o2) 0

The given vectors are Linearly Dependent and hence not a basis

(%o4) $\begin{bmatrix} 1 & \frac{1}{3} & 0 \\ 0 & 1 & \frac{9}{5} \\ 0 & 0 & 0 \end{bmatrix}$

The vectors $[1,\frac{1}{3},0]$, $[0,1,\frac{9}{5}]$ *are the basis of the subspace*

Dimension of the subspace is= 2

3)Find the basis and dimension of the subspace spanned by the vectors (2,-3,1), (3,0,1), (0,2,1), (1,1,1) of R^3

Lab4:**<u>Verifying whether a given transformation is linear</u>**

1) Verify whether $T: V_2(R) \longrightarrow V_2(R)$ defined by $T(x, y) = (x + y, y)$ is a linear transformation.

Program:

```
kill(all)$
T(x):=[x[1]+x[2], x[2]];
u[1]:[a,b]$
u[2]:[c,d]$
a1:T(u[1])+T(u[2]);
a2:radcan(T(u[1]+u[2]));
p1:factor(T(k*u[1]));
p2:factor(k*T(u[1]));
if (a1=a2 and p1=p2) then
print("The given mapping is a linear transformation")
else
print("The given mapping is not a linear transformation")$
```

Output:

```
(%o1) T(x):=[x[1]+x[2],x[2]]
(%o4) [d+c+b+a, d+b]
(%o5) [d+c+b+a, d+b]
(%o6) [(b+a)*k, b*k]
(%o7) [(b+a)*k, b*k]
"The given mapping is a linear transformation"
```

2) Verify whether $T: V_3(R) \longrightarrow V_3(R)$ defined by
$T(x, y, z) = (x^2 + xy,\ xy,\ yz)$ is a linear transformation.

Program:

```
kill(all)$
T(x):=[x[1]^2+x[1]*x[2],x[1]*x[2],x[2]*x[3]];
u[1]:[a,b,c]$
u[2]:[d,e,f]$
a1:T(u[1])+T(u[2]);
a2:radcan(T(u[1]+u[2]));
p1:factor(T(k*u[1]));
p2:factor(k*T(u[1]));
if (a1=a2 and p1=p2) then
print("The given mapping is a linear transformation")
```

```
else
print("The given mapping is not a linear transformation")$
```

Output:

```
(%o1) T(x):=[x[1]^2+x[1]*x[2],x[1]*x[2],x[2]*x[3]]
(%o4) [d*e+d^2+a*b+a^2, d*e+a*b, e*f+b*c]
(%o5) [(d+a)*e+d^2+(b+2*a)*d+a*b+a^2, (d+a)*e+b*d+a*b, (e+b)*f+c*e+b*c]
(%o6) [a*(b+a)*k^2, a*b*k^2, b*c*k^2]
(%o7) [a*(b+a)*k, a*b*k, b*c*k]
"The given mapping is not a linear transformation"
```

3) Verify whether $T: R^3 \longrightarrow R^2$ defined by $T(x, y, z) = (x + 2y - 3z,\ 4x - 5y + 6z)$ is a linear transformation.

4) Verify whether $T: V_2(R) \longrightarrow V_3(R)$ defined by $T(x, y) = (x + 3,\ 2y, x + y)$ is a linear transformation.

Lab4:**<u>Finding a linear transformation of a matrix.</u>**

1) Find the linear transformation for the matrix $\begin{bmatrix} -1 & 0 \\ 2 & 0 \\ 1 & 3 \end{bmatrix}$ with respect to the basis B1 = {(1,0), (2,-1)} and B2 = {(1,2,0), (0,-1,0),(1,-1,1)}

Program:

```
kill(all)$
A:matrix([-1,0],[2,0],[1,3]);
u[1]:[1,0]$
u[2]:[2,-1]$
v[1]:[1,2,0]$
v[2]:[0,-1,0]$
v[3]:[1,-1,1]$
for i:1 thru 2 do(
   T[i]:A[1,i]*v[1]+A[2,i]*v[2]+A[3,i]*v[3])$
for j:1 thru 2 do(
   eq[j]:u[1][j]*a+u[2][j]*b)$
soln:solve([eq[1]=x,eq[2]=y],[a,b])$
p:ev(a,soln)$
q:ev(b,soln)$
```

```
T[1]*p+T[2]*q;
T:radcan(T[1]*p+T[2]*q)$
print("The linear transformation T is T(x,y)=",T)$
```

Output:

$$\begin{bmatrix} -1 & 0 \\ 2 & 0 \\ 1 & 3 \end{bmatrix}$$

```
(%o12) [-3 y,3 y-5(2 y+x),x-y]
The linear transformation T is T(x,y)= [-3 y,-7 y-5 x,x-y]
```

2)Given the matrix of linear transformation of $\begin{bmatrix} 2 & 3 \\ 4 & -5 \end{bmatrix}$T, find T w.r.t the basis and B2 = {(1,-1), (1,1) }and B2 = {(1,0), (0,1)},

3)Find the linear transformation T w.r.t the standard basis, given the matrix of linear transformation A=$\begin{bmatrix} 1 & 0 & 1 \\ 2 & 1 & 1 \\ -1 & 1 & -2 \end{bmatrix}$

Finding a matrix of linear transformation

1) Find the matrix of linear transformation T:$R^2 \longrightarrow R^2$ defined by $T(x,y) = (2x + 3y, 4x - 5y)$ with respect to the standard basis.

Program:

```
kill(all)$
T(x):=[2*x[1]+3*x[2], 4*x[1]-5*x[2]];
u[1]:[1,0]$
u[2]:[0,1]$
v[1]:[1,0]$
v[2]:[0,1]$
for i:1 thru 2 do(
   eq[i]:v[1][i]*x+v[2][i]*y)$
for k:1 thru 2 do(
   soln:solve([eq[1]=T(u[k])[1],eq[2]=T(u[k])[2]],[x,y]),
      p[k]:ev(x,soln),
      q[k]:ev(y,soln))$
print("The matrix of linear transformation is A:")$
A:matrix([p[1],p[2]],[q[1],q[2]]);
```

Output:

T(x):=[2*x[1]+3*x[2], 4*x[1]-5*x[2]]

The matrix of linear transformation is A:

$$\begin{bmatrix} 2 & 3 \\ 4 & -5 \end{bmatrix}$$

2)Given $T(x,y,z) = \left(x - y + z, 2x + 3y - \frac{1}{2}z, x + y - 2z\right)$, find the matrix of T relative to the bases $B_1 = \{(-1,1,0), (5,-1,2), (1,2,1)\}$ and $B_2 = \{(1,1,0), (0,0,1), (1,5,2)\}$

Program:

```
kill(all)$
T(x):=[x[1]-x[2]+x[3],2*x[1]+3*x[2]-(1/2)*x[3],x[1]+x[2]-2*x[3]];
u[1]:[-1,1,0]$
u[2]:[5,-1,2]$
u[3]:[1,2,1]$
v[1]:[1,1,0]$
v[2]:[0,0,1]$
v[3]:[1,5,2]$
for i:1 thru 3 do(
   eq[i]:v[1][i]*x+v[2][i]*y+v[3][i]*z)$
for k:1 thru 3 do(
   soln:solve([eq[1]=T(u[k])[1],eq[2]=T(u[k])[2],eq[3]=T(u[k])[3]],[x,y,z]),
p[k]:ev(x,soln),
q[k]:ev(y,soln),
r[k]:ev(z,soln))$
print("The matrix of linear transformation is A:")$
A:matrix([p[1],p[2],p[3]],[q[1],q[2],q[3]],[r[1],r[2],r[3]]);
```

Output:

T(x):=[x[1]-x[2]+x[3] , 2*x[1]+3*x[2]+(-1/2)*x[3] , x[1]+x[2]+(-2)*x[3]]

The matrix of linear transformation is A:

$$\begin{bmatrix} \frac{-11}{4} & \frac{17}{2} & \frac{-15}{8} \\ \frac{-3}{2} & 1 & \frac{-11}{4} \\ \frac{3}{4} & \frac{-1}{2} & \frac{15}{8} \end{bmatrix}$$

3) Find the matrix of linear transformation T:$V_2(R) \longrightarrow V_3(R)$ defined by $T(x, y) = (-x + 2y, y, -3x + 3y)$ relative to the basis B_1={(1,2), (-2,1)} and B_2={(-1,0,2), (1,2,3),(1,-1,-1)}.

4)Find the matrix of linear transformation T:$V_3(R) \longrightarrow V_2(R)$ defined by relative to the basis $T(x_1, x_2, x_3) = (x_1 + x_2,\ x_2 + x_3)$
B1={(1,1,0), (1,0,1),(1,1,-1)} and B2={(2,-3), (1,4)}.

Lab5: **Verification of Rank-Nullity theorem.**

1)If linear transformation T:$R^3 \longrightarrow R^3$ is defined by $T(e_1)$=(1,-1,0), $T(e_2)$=(2,0,1) and $T(e_3)$=(1,1,1) then find range space, rank and nullity of T and hence verify rank-nullity theorem.

Program:

```
kill(all)$
d[V]:3;
disp("Dimension of domain space is",d[V]);
M:matrix([1,-1,0],[2,0,1],[1,1,1]);
r:rank(M);
disp("Dimension of the range space R(T) is",r)$
e:echelon(M);
disp("Range space R(T) is generated by",{e[1],e[2]})$
nullspace(M)$
n:nullity(M);
disp("Dimension of null space N(T) is",n)$
if d[V]=(r+n) then
disp("Rank-nullity theorem is verified")$
```

Output:

3

Dimension of domain space is:3

$$\begin{bmatrix} 1 & -1 & 0 \\ 2 & 0 & 1 \\ 1 & 1 & 1 \end{bmatrix}$$

2

Dimension of the range space R(T) is:2

$$\begin{bmatrix} 1 & -1 & 0 \\ 0 & 1 & \frac{1}{2} \\ 0 & 0 & 0 \end{bmatrix}$$

Range space R(T) is generated by: {[0,1, $\frac{1}{2}$],[1,-1,0]}

$$\text{span}\left\{\begin{bmatrix} -1 \\ -1 \\ 2 \end{bmatrix}\right\}$$

1

Dimension of null space N(T) is:1
Rank-nullity theorem is verified.

2) If linear transformation $T:R^3 \longrightarrow R^3$ is defined by $T(e_1)=(1,1,0)$, $T(e_2)=(0,1,1)$ and $T(e_3)=(1,2,1)$ then find range space, rank and nullity of T and hence verify rank-nullity theorem.

Program:

```
kill(all)$
d[V]:3;
disp("Dimension of domain space is:",d[V]);
M:matrix([1,1,0],[0,1,1],[1,2,1]);
r:rank(M);
disp("Dimension of the range space R(T) is:",r)$
e:echelon(M);
disp("Range space R(T) is generated by",{e[1],e[2]})$
nullspace(M);
n:nullity(M);
disp("Dimension of null space N(T) is:",n)$
if d[V]=(r+n) then
disp("Rank-nullity theorem is verified")$
```

Output:

3

Dimension of domain space is:3

$$\begin{bmatrix} 1 & 1 & 0 \\ 0 & 1 & 1 \\ 1 & 2 & 1 \end{bmatrix}$$

2

Dimension of the range space R(T) is:2

$$\begin{bmatrix}1 & 1 & 0\\0 & 1 & 1\\0 & 0 & 0\end{bmatrix}$$

Range space R(T) is generated by:{[0,1,1],[1,1,0]}

$$\text{span}\left\{\begin{bmatrix}-1\\1\\-1\end{bmatrix}\right\}$$

1

Dimension of null space N(T) is:1

"Rank-nullity theorem is verified."

3) If linear transformation $T:R^3 \longrightarrow R^3$ is defined by $T(e_1)=(1,1,0)$, $T(e_2)=(1,0,1)$ and $T(e_3)=(0,1,1)$ then find range space, rank and nullity of T and hence verify rank-nullity theorem.

4) If linear transformation $T:R^3 \longrightarrow R^3$ is defined by $T(e_1)=(1,1,2)$,

$T(e_2)=(1,-1,0)$ and $T(e_3)=(0,0,1)$ then find range space, rank and nullity of T and hence verify rank-nullity theorem.

Lab6:**<u>Solutions to problems on total differential equations.</u>**

Verify the condition of integrability

$$(2x^2 + 2xy + 2xz^2 + 1)dx + dy + 2zdz = 0$$

Program

```
kill(all)$
load("lrats")$
P:2*x^2+2*x*y+2*x*z^2+1$
Q:1$
R:2*z$
eqn:P*'del(x)+Q*'del(y)+R*'del(z);
I:P*(diff(Q,z)-diff(R,y))+Q*(diff(R,x)-diff(P,z))+R*(diff(P,y)-diff(Q,x))$
if I#0 then disp("The given equation is not integrable,
hence there is no solution")
else
```

```
disp ("The given equation is integrable")$
```

Output:

```
del(y)+2*z*del(z)+(2*x*z^2+2*x*y+2*x^2+1)*del(x)
"The given equation is integrable"
```

Problems

2) $2yzdx + zxdy - xy(1 + z)dz = 0$

3) $(y^2 + yz)dx + (zx + z^2)dy + (y^2 - xy)dz = 0$

Lab7:**<u>Solutions to the problems on different types of partial differential equations (Type1 &Type2)</u>**

Type 1:

Equations of the form f(p,q)=0

1)Solve: pq=1.

Program

```
kill(all)$
eqn:(p*q=1);
z:a*x+b*y+c;
h:subst([p=a,q=b],eqn);
solve(h,a);
h1:subst(%,z)$
disp("The required solution is:",h1)$
```

Output:

```
p*q=1
[a=1/b]
"The required solution is:"
b*y+x/b+c
```

Solve

2) $p^2+q^2 =3$,
3) $pq + p + z = 0$,

Type 2: Equations of the form f(p,q,z)=0

Solve: $p(1-q^2)=q(1-z)$

Program:

```
kill(all)$
eqn:p*(1-q^2)=q*(1-z);
eqn1:subst([p='diff(z,u),q=a*'diff(z,u)],eqn)$
h1:solve(eqn1,'diff(z,u))$
ode2(h1[1],z,u)$
h2:subst(u=x+a*y,%)$
disp("solution is:",h2)$
```

Output:

```
p*(1-q^2)=q*(1-z)
" solution is:"
-2*sqrt(a*z-a+1)=a*y+x+%c
```

2)Solve: $p(1+q)=qz$

3) Solve: p=qz

Lab8:**<u>Solutions to the problems on different types of partial differential equations (Type3 &Type4)</u>**

Type 3:Equations of the form $f_1(p,x)=f_2(q,y)$.

1)Solve: p+x=q+y

Program:

```
kill(all)$
eqn:p+x=q+y;
d:(p+q*'diff(y,x))$
r1:lhs(eqn)=k$
r2:rhs(eqn)=k$
h1:solve(r1,p)$
h2:solve(r2,q)$
subst(h1,d)$
A:subst(h2,%)$
ode2((A),y,x)$
z: rhs(%)-lhs(%)$
disp("the solutionis z=",z)$
```

Output:

```
x+p=y+q
"the solutionis z="
(y^2-2*k*y)/2+(x^2-2*k*x)/2-%c
```

2)Solve: p-cos(x)=cos(y)/q
3)Solve:$pe^{x} = qe^{y}$.
Type 4:Clariaut's equation
Solve: $z = px + qy + pq$
Program:
```
kill(all)$
eqn:z=p*x+q*y+p*q;
soln:subst([p=a,q=b],%)$
disp("solution is s:", soln)$
```
Output:
```
z=q*y+p*x+p*q
"solution is "
z=b*y+a*x+a*b
```
2) Solve $z = px + qy + \sqrt{\frac{pq}{p+q}}$
3) Solve $z = px + qy + \log(pq)$.

Lab9:**<u>Solving second order linear partial differential equations in two variables with constant Co-efficients.</u>**

To find the complimentary function
Solve:$\left(D^2 + 3DD' + 2D'^2\right) z = 0$.

Program:
```
kill(all)$
F(D,D1)*z=0$
F(D,D1):=D^2+3*D*(D1)+2*(D1)^2;
ae:F(m,1)$
k:allroots(ae);
k1:rhs(k[1])$
k2:rhs(k[2])$
a1:f(y+k1*x)+g(y+k2*x)$
a2:f(y+k1*x)+x*g(y+k2*x)$
if k1#k2 then
 disp("solution is",z=a1)
else disp("solution is ",z=a2)$
```
Output:
```
F(D,D1):=D^2+3*D*D1+2*D1^2=0
```

```
[m=-1.0,m=-2.0]
"solution is"
z=f(y-1.0*x)+g(y-2.0*x)
```

Solve the following equations

2) $\left(D^2 - 4DD' + 4D'^2\right)z = 0.$

3) $\left(2D^2 + 5DD' + 2D'^2\right)z = 0.$

4) $(D^2 - DD')z = 0$

Lab10:**Solving second order linear partial differential equations in two variables with constant Co-efficients.**

Solve $\left(D^2 - DD' - 2D'^2\right)z = e^{x+2y}$

Program:

```
kill(all)$
F(D,D1)*z=f(x,y)$
F(D,D1):=D^2-D*D1-2*D1^2;
f(x,y):=%e^(x+2*y);
ae:F(m,1)$
h:allroots(ae=0)$
h1:rhs(h[1])$
h2:rhs(h[2])$
cf1:f1(y+h1*x)+g1(y+h2*x)$
cf2:f1(y+h1*x)+x*g1(y+h2*x)$
if h1#h2 then
(CF:cf1)
else
(CF:cf2)$
disp(CF)$
I1:integrate(f(x,c-h1*x),x)$
f3(x,y):=ratsimp(subst([c=y+h1*x],I1))$
integrate(f3(x,c-h2*x),x)$
PI:ratsimp(subst([c=y+h2*x],%));
z:CF+PI;
```

Output:

```
F(D,D1):=D^2-D*D1+(-2)*D1^2
 f(x,y):=%e^(x+2*y)
g1(y+2.0*x)+f1(y-1.0*x)
rat: replaced -4.0 by -4/1 = -4.0
rat: replaced 0.3333333333333333 by 1/3 = 0.3333333333333333
rat: replaced 4.0 by 4/1 = 4.0
 -%e^(2*y+x)/9
 g1(y+2.0*x)+f1(y-1.0*x)-%e^(2*y+x)/9
```

Solve the following equations

2) $\left(D^2 - 5DD' + 4D'^2\right)z = \cos(2x + 3y)$

3) $\left(D^2 + 3DD' + 2D'^2\right)z = x + y$

4) $\left(D^2 - DD' - 2D'^2\right)z = (y - 1)e^x$

5) $(D^2 - 2DD' + D'^2)z = 12xy$

6) $(D^2 - 2DD' + D'^2)z = \sin(x + y)$

Lab11:**Solution of one dimensional heat equation using Fourier series with Dirichlet condition.**

Solve $\frac{\partial u}{\partial t} = 16\frac{\partial^2 u}{\partial x^2}$ *subject to the condition* $i) u(0,t) = 0, u(1,t) = 0$ *for all* t, $ii) u(x,0) = x^2 - x, 0 \le x \le 1$.

Program

```
kill(all)$
load("fourie")$
load("lrats")$
g(x,t):=diff(u(x,t),t)=16*diff(u(x,t),x,2);
declare(n,integer)$
assume(n>0,c>0,x>0,t>0,a>0)$
u(x,t):=X(x)*T(t)$
F(x,t):=g(x,t)/(16*(u(x,t)))$
x1:ode2(rhs(F(x,t))=-a^2,X(x),x)$
```

```
define(X(x),rhs(x1))$
t1:ode2(lhs(F(x,t))=-a^2,T(t),t)$
define(T(t),rhs(t1))$
disp("applying the condition X(0)=0")$
if at(X(x),x=0)=%k2 then %k2:0 else %k1:0$
disp("X(x)=",X(x))$
if at(X(x),x=1)#0 then
   a:n*%pi
else a:0$
disp("X(x)=",X(x))$
u(x,t)$
f(x):=x^2-x;
b[n]:2*integrate(f(x)*sin(%pi*n*x),x,0,1);
C:lratsubst(%c*%k1=(b[n]),u(x,t))$
u1:sum(C,n,1,inf)$
disp("solution is u(x,t)=",u1)$
```

Output:

(%o3) g(x,t):=diff(u(x,t),t)=16 diff(u(x,t),x,2)

applying the condition X(0)=0

X(x)=

$\%k1 \sin(a\, x)$

X(x)=

$\%k1 \sin(\pi\, n\, x)$

(%o18) $f(x) := x^2 - x$

(%o19) $2\left(\frac{2(-1)^n}{\pi^3 n^3} - \frac{2}{\pi^3 n^3}\right)$

solution is u(x,t)=

$$\frac{\sum_{n=1}^{\infty} \frac{(4(-1)^n - 4)\, \%e^{-16\pi^2 n^2 t} \sin(\pi\, n\, x)}{n^3}}{\pi^3}$$

Lab12: **<u>Solution of one dimensional wave equation using Fourier series with Dirichlet condition.</u>**

Solve $\frac{\partial^2 u}{\partial t^2} = \frac{\partial^2 u}{\partial x^2}$ *subject to the condition* $i)\ u(0,t) = 0, u(l,t) = 0\ for\ all\ t,$ $ii)\ u(x,0) = 0 - x,\ \left(\frac{\partial u}{\partial t}\right)_{t=0} = lx - x^2,\ 0 \le x \le l.$

Program:

```
kill(all)$
load("fourie")$
load("lrats")$
g(x,t):=diff(u(x,t),t,2)=diff(u(x,t),x,2);
assume(n>0,c>0,x>0,t>0,a>0,l>0)$
u(x,t):=X(x)*T(t)$
F(x,t):=g(x,t)/(u(x,t))$
x1:ode2(rhs(F(x,t))=-a^2,X(x),x)$
define(X(x),rhs(x1))$
t2:ode2(lhs(F(x,t))=-a^2,T(t),t)$
t1:subst ([%k1=%k3 ,%k2=%k4], %)$
t2:define(T(t),rhs(t1))$
disp("applying the condition X(0)=0")$
if at(X(x),x=0)=%k2 then %k2:0 else %k1:0$
disp("X(x)=",X(x))$
if at(X(x),x=l)#0 then a:n*%pi/l else a:0$
disp("X(x)=",X(x))$
u1:ratsimp(u(x,t))$
f(x):=0$
2/l*integrate(f(x)*sin(n*%pi*x/l),x,0,l)$
B[n]:foursimp(%);
g(x):=l*x-x^2;
2/(n*%pi)*integrate(g(x)*sin(n*%pi*x/l),x,0,l)$
A[n]:foursimp(%);
u2:lratsubst([%k1*%k3=A[n], %k1*%k4=B[n]],u1)$
u3:sum(u2,n,1,inf)$
disp("solution is u(x,t)=",u3)$
```

Output:

(%o3) g(x,t):=diff(u(x,t),t,2)=diff(u(x,t),x,2)

applying the condition X(0)=0

$X(x)=$

$\%k1\sin(a\,x)$

$X(x)=$

$\%k1\sin\left(\frac{\pi\,n\,x}{l}\right)$

(%o20) 0

(%o21) $g(x):=l\,x-x^2$

(%o23) $-\frac{4\,l^3\left((-1)^n-1\right)}{\pi^4\,n^4}$

solution is u(x,t)=

$$-\frac{\sum_{n=1}^{\infty}\frac{(4\,l^3(-1)^n-4\,l^3)\sin\left(\frac{\pi\,n\,t}{l}\right)\sin\left(\frac{\pi\,n\,x}{l}\right)}{n^4}}{\pi^4}$$

****